// SPDX-License-Identifier: MIT
pragma solidity ^0.8.18;

// TODO Change this before deployment 
// they can be changed later but will cost gas

// 10% boost for wood it mmeans if base apy was 300 then it will be 330
uint256 constant WOOD_BOOST_PERCENT = 10; 

// 25% boost for iron it mmeans if base apy was 300 then it will be 375
uint256 constant IRON_BOOST_PERCENT = 25; 

// 50% boost for gold it mmeans if base apy was 300 then it will be 450
uint256 constant GOLD_BOOST_PERCENT = 50; 

// 100% boost for diamond it mmeans if base apy was 300 then it will be 600
uint256 constant DIAMOND_BOOST_PERCENT = 100;

// 
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)
/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// 
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)
/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// 
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)
/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// 
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)
/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}

// 
// OpenZeppelin Contracts (last updated v4.8.0) (utils/Address.sol)
/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResultFromTarget(target, success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verify that a low level call to smart-contract was successful, and revert (either by bubbling
     * the revert reason or using the provided one) in case of unsuccessful call or if target was not a contract.
     *
     * _Available since v4.8._
     */
    function verifyCallResultFromTarget(
        address target,
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        if (success) {
            if (returndata.length == 0) {
                // only check isContract if the call was successful and the return data is empty
                // otherwise we already know that it was a contract
                require(isContract(target), "Address: call to non-contract");
            }
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    /**
     * @dev Tool to verify that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason or using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            _revert(returndata, errorMessage);
        }
    }

    function _revert(bytes memory returndata, string memory errorMessage) private pure {
        // Look for revert reason and bubble it up if present
        if (returndata.length > 0) {
            // The easiest way to bubble the revert reason is using memory via assembly
            /// @solidity memory-safe-assembly
            assembly {
                let returndata_size := mload(returndata)
                revert(add(32, returndata), returndata_size)
            }
        } else {
            revert(errorMessage);
        }
    }
}

// 
// OpenZeppelin Contracts (last updated v4.8.0) (token/ERC20/utils/SafeERC20.sol)
/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address-functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

interface IPixelShibellasERC721 {
  function approve ( address to, uint256 tokenId ) external;
  function balanceOf ( address owner ) external view returns ( uint256 );
  function baseURI (  ) external view returns ( string memory );
  function getApproved ( uint256 tokenId ) external view returns ( address );
  function isApprovedForAll ( address owner, address operator ) external view returns ( bool );
  function maxSupply (  ) external view returns ( uint256 );
  function mintPixelShibellas ( uint256 _numToMint ) external;
  function name (  ) external view returns ( string memory );
  function owner (  ) external view returns ( address );
  function ownerOf ( uint256 tokenId ) external view returns ( address );
  function rarity ( uint256 tokenId ) external view returns ( uint256 );
  function renounceOwnership (  ) external;
  function safeTransferFrom ( address from, address to, uint256 tokenId ) external;
  function safeTransferFrom ( address from, address to, uint256 tokenId, bytes calldata _data  ) external;
  function setApprovalForAll ( address operator, bool approved ) external;
  function supportsInterface ( bytes4 interfaceId ) external view returns ( bool );
  function symbol (  ) external view returns ( string memory );
  function teamMint ( address to, uint256 tokenId ) external;
  function tokenURI ( uint256 tokenId ) external view returns ( string memory );
  function totalSupply (  ) external view returns ( uint256 );
  function transferFrom ( address from, address to, uint256 tokenId ) external;
  function transferOwnership ( address newOwner ) external;
  function withdraw (  ) external;
}

// 
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC721/IERC721Receiver.sol)
/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721Receiver.onERC721Received.selector`.
     */
    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes calldata data
    ) external returns (bytes4);
}

// 
// import "hardhat/console.sol";

struct PoolInfo {
    uint256 totalStaked;
    uint256 totalBoostedStake;
    uint256 totalWithdrawFee;
    uint256 totalWithdrawFeeClaimed;
    uint256 rewardPerShare;
    uint256 lastRewardTime;
    uint256 apy;
    uint256 rewardsPerSecond;
    uint256 startTime;
    uint256 endTime;
    uint256 totalRewards;
    uint256 withdrawFee;
    uint256[] rarityBoosts;
}

struct UserInfo {
    uint256 amount;
    uint256 boostedAmount;
    uint256 rewardDebt;
    uint256 nftId;
    uint256 currentBoost;
    uint256 totalRewardsClaimed;
    uint256 totalWithdrawFee;
}

contract ShibellasStaking is Ownable, IERC721Receiver {
    using SafeERC20 for IERC20;

    uint256 DENOMINATOR = 100;

    uint256 constant ACC_PER_SHARE_PRECISION = 1e32;

    IERC20 public immutable token;
    IPixelShibellasERC721 public immutable pixelshibellasERC721;

    PoolInfo public poolInfo;

    uint256 _rewardsSinceLastUpdate;
    uint256 _lastUpdateTime;

    mapping(address => UserInfo) public userInfo;

    event Deposit(address indexed user, uint256 amount);
    event Claimed(address indexed user, uint256 amount);
    event Withdraw(address indexed user, uint256 amount, uint256 fee);
    event EmergencyWithdraw(address indexed user, uint256 amount);

    event DepositNFT(
        address indexed nftAddress,
        uint256 indexed nftId,
        address indexed from,
        address operator
    );

    constructor(IERC20 _token, IPixelShibellasERC721 _pixelshibellasERC721) {
        token = _token;
        poolInfo.rarityBoosts = [
            DIAMOND_BOOST_PERCENT,
            GOLD_BOOST_PERCENT,
            IRON_BOOST_PERCENT,
            WOOD_BOOST_PERCENT
        ];

        pixelshibellasERC721 = _pixelshibellasERC721;
    }

    function setNFTRarities(
        uint256 _wood,
        uint256 _iron,
        uint256 _gold,
        uint256 _diamond
    ) external onlyOwner {
        poolInfo.rarityBoosts = [_diamond, _gold, _iron, _wood];
    }

    function getRarityBoosts() external view returns (uint256[] memory) {
        return poolInfo.rarityBoosts;
    }

    function onERC721Received(
        address operator,
        address from,
        uint256 tokenId,
        bytes memory
    ) public override returns (bytes4) {
        require(
            msg.sender == address(pixelshibellasERC721),
            "onERC721Received: Only accepts PixelshibellasERC721 tokens"
        );
        require(operator == address(this), "onERC721Received: Only as boost");

        // Emit an event to notify the recipient of the deposited token
        emit DepositNFT(msg.sender, tokenId, from, operator);

        return IERC721Receiver.onERC721Received.selector;
    }

    function boostDeposit(uint256 _nftId) external {
        UserInfo storage user = userInfo[msg.sender];

        require(user.nftId == 0, "boostDeposit: already boosted");

        pixelshibellasERC721.safeTransferFrom(msg.sender, address(this), _nftId);

        user.nftId = _nftId;

        uint256 rarity = pixelshibellasERC721.rarity(_nftId);

        updatePool();

        uint256 userAmount = user.amount;
        // claim rewards before boost
        if (userAmount > 0) {
            uint256 pendingNow = (user.boostedAmount *
                poolInfo.rewardPerShare) /
                ACC_PER_SHARE_PRECISION -
                user.rewardDebt;
            if (pendingNow > 0) {
                token.safeTransfer(msg.sender, pendingNow);
                user.totalRewardsClaimed += pendingNow;
                emit Claimed(msg.sender, pendingNow);
            }
        }
        user.currentBoost = poolInfo.rarityBoosts[rarity - 1];
        uint256 oldBoostedAmount = user.boostedAmount;
        user.boostedAmount =
            userAmount +
            (userAmount * user.currentBoost) /
            DENOMINATOR;
        poolInfo.totalBoostedStake =
            poolInfo.totalBoostedStake -
            oldBoostedAmount +
            user.boostedAmount;

        user.rewardDebt =
            (user.boostedAmount * poolInfo.rewardPerShare) /
            ACC_PER_SHARE_PRECISION;

        _updateRewardsPerSecond();
    }

    function removeBoost() external {
        UserInfo storage user = userInfo[msg.sender];

        require(user.nftId != 0, "removeBoost: not boosted");

        updatePool();

        pixelshibellasERC721.safeTransferFrom(
            address(this),
            msg.sender,
            user.nftId
        );

        uint256 userAmount = user.amount;
        uint256 oldBoostedAmount = user.boostedAmount;

        // claim rewards before boost
        if (userAmount > 0) {
            uint256 pendingNow = (oldBoostedAmount * poolInfo.rewardPerShare) /
                ACC_PER_SHARE_PRECISION -
                user.rewardDebt;
            if (pendingNow > 0) {
                token.safeTransfer(msg.sender, pendingNow);
                user.totalRewardsClaimed += pendingNow;
                emit Claimed(msg.sender, pendingNow);
            }
        }

        user.boostedAmount = userAmount;
        poolInfo.totalBoostedStake =
            poolInfo.totalBoostedStake -
            oldBoostedAmount +
            user.boostedAmount;

        user.nftId = 0;
        user.currentBoost = 0;

        user.rewardDebt =
            (user.boostedAmount * poolInfo.rewardPerShare) /
            ACC_PER_SHARE_PRECISION;
        _updateRewardsPerSecond();
    }

    modifier onlyWhenActive() {
        require(
            poolInfo.startTime > 0 &&
                block.timestamp >= poolInfo.startTime &&
                block.timestamp <= poolInfo.endTime,
            "onlyWhenActive: not active"
        );
        _;
    }

    function initStaking(
        uint256 _startTime,
        uint256 _apy,
        uint256 _withdrawFee,
        uint256 _rewardsAmount
    ) external onlyOwner {
        require(poolInfo.startTime == 0, "initStaking: already initialized");
        require(
            _startTime >= block.timestamp,
            "initStaking: invalid start time"
        );

        poolInfo.startTime = _startTime;
        poolInfo.endTime = type(uint256).max;
        poolInfo.withdrawFee = _withdrawFee;
        poolInfo.lastRewardTime = _startTime;
        _lastUpdateTime = _startTime;

        uint256 netAmount = _amountSafeERC20TransferFrom(
            msg.sender,
            _rewardsAmount
        );
        require(netAmount == _rewardsAmount, "initStaking: tax enabled");

        poolInfo.totalRewards = netAmount;
        poolInfo.apy = _apy;
    }

    function updatePool() public {
        uint256 lastTime = block.timestamp > poolInfo.endTime
            ? poolInfo.endTime
            : block.timestamp;

        if (lastTime <= poolInfo.lastRewardTime) {
            return;
        }

        if (poolInfo.totalBoostedStake == 0) {
            poolInfo.lastRewardTime = lastTime;
            return;
        }

        uint256 timePassed = lastTime - poolInfo.lastRewardTime;
        uint256 reward = (timePassed * poolInfo.rewardsPerSecond);

        poolInfo.rewardPerShare =
            poolInfo.rewardPerShare +
            ((reward * ACC_PER_SHARE_PRECISION) / poolInfo.totalBoostedStake);

        poolInfo.lastRewardTime = lastTime;
    }

    function deposit(uint256 _amount) external onlyWhenActive {
        uint256 netAmount = _amountSafeERC20TransferFrom(msg.sender, _amount);

        // uint256 netAmount = netTransfer - fee;

        UserInfo storage user = userInfo[msg.sender];

        updatePool();

        // claim pending rewards
        if (user.amount > 0) {
            uint256 pendingNow = (user.boostedAmount *
                poolInfo.rewardPerShare) /
                ACC_PER_SHARE_PRECISION -
                user.rewardDebt;
            if (pendingNow > 0) {
                token.safeTransfer(msg.sender, pendingNow);
                user.totalRewardsClaimed += pendingNow;
                emit Claimed(msg.sender, pendingNow);
            }
        }
        uint256 _boostedAmount = netAmount +
            (netAmount * user.currentBoost) /
            DENOMINATOR;

        poolInfo.totalStaked += netAmount;
        user.amount += netAmount;
        user.boostedAmount += _boostedAmount;
        poolInfo.totalBoostedStake += _boostedAmount;
        user.rewardDebt =
            (user.boostedAmount * poolInfo.rewardPerShare) /
            ACC_PER_SHARE_PRECISION;

        _updateRewardsPerSecond();

        emit Deposit(msg.sender, netAmount);
    }

    function withdraw(uint256 _amount) external {
        UserInfo storage user = userInfo[msg.sender];

        require(user.amount >= _amount, "withdraw: not enough balance");

        updatePool();

        uint256 pendingNow = (user.boostedAmount * poolInfo.rewardPerShare) /
            ACC_PER_SHARE_PRECISION -
            user.rewardDebt;
        if (pendingNow > 0) {
            token.safeTransfer(msg.sender, pendingNow);
            user.totalRewardsClaimed += pendingNow;
            emit Claimed(msg.sender, pendingNow);
        }

        if (_amount > 0) {
            uint256 fee = (_amount * poolInfo.withdrawFee) / DENOMINATOR;
            poolInfo.totalWithdrawFee += fee;
            user.totalWithdrawFee += fee;
            uint256 netAmount = _amount - fee;
            token.safeTransfer(msg.sender, netAmount);

            emit Withdraw(msg.sender, _amount, fee);
            uint256 _boostedAmount = _amount +
                (_amount * user.currentBoost) /
                DENOMINATOR;
            poolInfo.totalStaked -= _amount;
            user.amount -= _amount;
            user.boostedAmount -= _boostedAmount;
            poolInfo.totalBoostedStake -= _boostedAmount;
            _updateRewardsPerSecond();
        }

        user.rewardDebt =
            (user.boostedAmount * poolInfo.rewardPerShare) /
            ACC_PER_SHARE_PRECISION;
    }

    function pending(address _user) external view returns (uint256) {
        UserInfo storage user = userInfo[_user];

        uint256 _rewardPerShare = poolInfo.rewardPerShare;
        uint256 lastTime = block.timestamp < poolInfo.endTime
            ? block.timestamp
            : poolInfo.endTime;
        if (
            lastTime > poolInfo.lastRewardTime &&
            poolInfo.totalBoostedStake != 0
        ) {
            uint256 timePassed = lastTime - poolInfo.lastRewardTime;
            uint256 reward = (timePassed * poolInfo.rewardsPerSecond);
            _rewardPerShare =
                _rewardPerShare +
                ((reward * ACC_PER_SHARE_PRECISION) /
                    poolInfo.totalBoostedStake);
        }

        return
            (user.boostedAmount * _rewardPerShare) /
            ACC_PER_SHARE_PRECISION -
            user.rewardDebt;
    }

    function _amountSafeERC20TransferFrom(
        address _from,
        uint256 _amount
    ) internal returns (uint256) {
        uint256 balanceBefore = token.balanceOf(address(this));
        token.transferFrom(_from, address(this), _amount);
        uint256 balanceAfter = token.balanceOf(address(this));

        return balanceAfter - balanceBefore;
    }

    function _updateRewardsPerSecond() internal {
        if (
            block.timestamp > poolInfo.endTime ||
            block.timestamp < poolInfo.startTime
        ) {
            return;
        }

        uint256 newRewardsPerSecond = (poolInfo.totalBoostedStake *
            poolInfo.apy) /
            365 days /
            100;

        uint256 totalRewardsTillNow = _rewardsSinceLastUpdate +
            (block.timestamp - _lastUpdateTime) *
            poolInfo.rewardsPerSecond;

        if (newRewardsPerSecond == 0) {
            // TODO: handle this case
            poolInfo.endTime = type(uint256).max;
        } else {
            uint256 totalRewardsLeft = poolInfo.totalRewards -
                totalRewardsTillNow;

            uint256 newSecondsLeft = totalRewardsLeft / newRewardsPerSecond;

            poolInfo.endTime = block.timestamp + newSecondsLeft;
        }
        _lastUpdateTime = block.timestamp;
        _rewardsSinceLastUpdate = totalRewardsTillNow;

        poolInfo.rewardsPerSecond = newRewardsPerSecond;
    }

    function setWithdrawFee(uint256 _fee) external onlyOwner {
        require(_fee <= DENOMINATOR, "setWithdrawFee: invalid fee");
        poolInfo.withdrawFee = _fee;
    }

    function setApy(uint256 _apy) external onlyOwner {
        updatePool();
        poolInfo.apy = _apy;
        _updateRewardsPerSecond();
    }

    function setStartTime(uint256 _startTime) external onlyOwner {
        require(
            block.timestamp < poolInfo.startTime,
            "setStartTime: already started"
        );
        // _startTime has to be in the future
        require(
            _startTime > block.timestamp,
            "setStartTime: invalid start time"
        );

        poolInfo.startTime = _startTime;
        poolInfo.lastRewardTime = _startTime;

        _lastUpdateTime = _startTime;
    }

    function depositRewards(uint256 _amount) external onlyOwner {
        uint256 netTransfer = _amountSafeERC20TransferFrom(msg.sender, _amount);
        require(netTransfer == _amount, "depositRewards: tax detected");
        poolInfo.totalRewards += netTransfer;
        _updateRewardsPerSecond();
    }

    function removeRewards(uint256 _amount) external onlyOwner {
        uint256 lastTime = block.timestamp < poolInfo.endTime
            ? block.timestamp
            : poolInfo.endTime;
        uint256 rewardsTillNow = _rewardsSinceLastUpdate +
            (lastTime - _lastUpdateTime) *
            poolInfo.rewardsPerSecond;

        uint256 rewardsLeft = poolInfo.totalRewards - rewardsTillNow;
        require(rewardsLeft >= _amount, "removeRewards: not enough rewards");
        poolInfo.totalRewards -= _amount;
        token.safeTransfer(msg.sender, _amount);
        _updateRewardsPerSecond();
    }

    function endTheFarmNow() external onlyOwner {
        uint256 lastTime = block.timestamp < poolInfo.endTime
            ? block.timestamp
            : poolInfo.endTime;
        uint256 rewardsTillNow = _rewardsSinceLastUpdate +
            (lastTime - _lastUpdateTime) *
            poolInfo.rewardsPerSecond;

        uint256 rewardsLeft = poolInfo.totalRewards - rewardsTillNow;

        if (rewardsLeft > 0) {
            token.safeTransfer(msg.sender, rewardsLeft);
        }

        poolInfo.totalRewards -= rewardsLeft;
        _updateRewardsPerSecond();
    }

    function withdrawFeesCollected() external onlyOwner {
        uint256 unclaimed = poolInfo.totalWithdrawFee -
            poolInfo.totalWithdrawFeeClaimed;
        require(unclaimed > 0, "withdrawFeesCollected: no fees to withdraw");
        token.safeTransfer(msg.sender, unclaimed);
        poolInfo.totalWithdrawFeeClaimed += unclaimed;
    }

    function emergencyWithdraw() external {
        require(block.timestamp > poolInfo.endTime, "only after end of farm");
        UserInfo storage user = userInfo[msg.sender];

        uint256 fee = (user.amount * poolInfo.withdrawFee) / DENOMINATOR;

        if (fee > 0) {
            poolInfo.totalWithdrawFee += fee;
        }

        token.safeTransfer(msg.sender, user.amount - fee);
        emit EmergencyWithdraw(msg.sender, user.amount);

        poolInfo.totalStaked -= user.amount;
        poolInfo.totalBoostedStake -= user.boostedAmount;
        user.amount = 0;
        user.boostedAmount = 0;
        user.rewardDebt = 0;

        // _updateRewardsPerSecond();
    }

    function getRewardsLeft() external view returns (uint256 rewardsLeft) {
        uint256 lastTime = block.timestamp < poolInfo.endTime
            ? block.timestamp
            : poolInfo.endTime;
        uint256 rewardsTillNow = _rewardsSinceLastUpdate +
            (lastTime - _lastUpdateTime) *
            poolInfo.rewardsPerSecond;

        rewardsLeft = poolInfo.totalRewards - rewardsTillNow;
    }

    function recoverERC20(address _token, uint256 _amount) external onlyOwner {
        require(
            _token != address(token) ||
                (poolInfo.endTime < block.timestamp &&
                    poolInfo.totalStaked == 0),
            "recoverERC20: cannot recover token before end of farm"
        );
        IERC20(_token).safeTransfer(msg.sender, _amount);
    }
}